
********************************************************************************
****** The Effects of WWII on Economic and Health Outcomes across Europe *******
********************************************************************************
* Authors: Iris Kesternich, Bettina Siflinger, James P. Smith, Joachim Winter
* Review of Economics and Statistics, 2014
********************************************************************************
* DOFILE: FIGURE 1, CASUALTIES
********************************************************************************


clear
clear matrix
clear mata
set more off


*** define the path to the directory containing the data files here 
*** or leave "." if the do files are in the current directory 

global datapath "."

*** define the path to the directory to which figures are saved here 
*** or leave "." if they should be saved to the current directory 

global graphpath  "."


insheet using "$datapath\external\txt\WWII_casualties.txt"
sort country
save "$datapath\external\dta\WWII_casualties.dta", replace


*** proportions casualties
gen prop_civ = civilian_deaths / tot_pop1939 
gen prop_hol = holocaust / tot_pop1939
gen prop_mil = military_deaths / tot_pop1939



recode country (11=1)(12=2)(16=3) (28=4)(19=5)(29=6)(23=7)(17=8)(14=9)(18=10)(15=11)(13=12)(20=13)(32=14)(30=15)(31=16)
gen country2 = country + 0.2
gen country3 = country2 + 0.2
gen country4 = country3 + 0.2


*** Figure 1a: WWII casualties as percentage of population 
	twoway (bar prop_deaths country, barw(0.2) bcolor(*.7))(bar prop_civ country2, barw(0.2) bcolor(*.7)) (bar prop_hol country3, barw(0.2) bcolor(*.7)) ///
	(bar prop_mil country4, barw(0.2) bcolor(*.7)), ///
	legend( order(1 "total deaths/pop 39" 2 "civilian deaths/pop 39" 3 "holocaust deaths/pop 39" 4 "military deaths/pop 39")) ///
	xlabel(1 "AUS" 2 "GER" 3 "ITA" 4 "CZE" 5 "GRE" 6 "POL" 7 "BEL" 8 "FRA" 9 "NED" 10 "DEN" 11 "ESP" 12 "SWE" 13 "SUI" 14 "USSR" 15 "GBR" 16 "USA", val angle(315)) ///
	yscale(range(0 .2)) ylabel(.1 "10%" .2 "20%") ytitle(WWII casualties) xtitle(country) xsize(8) ysize(5) ///
	note(Own calculations. Based on F.W. Putzger (1963); van Mourik (1978); R. Overmans (1999))
	graph export "$graphpath\casualties_europe_prop_Soviet_UK_US.png", width(600) replace


*** Figure 1b: number of WWII casualties 
	twoway (bar sum_deaths country, barw(0.2) bcolor(*.7))(bar civilian_deaths country2, barw(0.2) bcolor(*.7)) (bar holocaust country3, barw(0.2) bcolor(*.7)) ///
	(bar military_deaths country4, barw(0.2) bcolor(*.7)), ///
	legend( order(1 "total deaths" 2 "civilian deaths" 3 "holocaust deaths" 4 "military deaths")) ///
	xlabel(1 "AUS" 2 "GER" 3 "ITA" 4 "CZE" 5 "GRE" 6 "POL" 7 "BEL" 8 "FRA" 9 "NED" 10 "DEN" 11 "ESP" 12 "SWE" 13 "SUI" 14 "USSR" 15 "GBR" 16 "USA", val angle(315)) ///
	yscale(range(0 22000000)) ylabel(5000000 "5 Mio" 10000000 "10 Mio" 15000000 "15 Mio" 20000000 "20 Mio" ) ///
	ytitle(Total number of WWII casualties) xtitle(country) xsize(8) ysize(5) ///
	note(Own calculations. Based on F.W. Putzger (1963); van Mourik (1978); R. Overmans (1999))
	graph export "$graphpath\casualties_europe_total_Soviet_UK_US.png", replace





